

MAX3273 I/O Model

SPICE I/O Macromodels aid in understanding signal integrity issues in electronic systems. Most of Maxim's High Frequency/Fiber Communication ICs utilize input and output (I/O) circuits with Current Mode Logic (CML), Positive Emitter Coupled Logic (PECL), and Low Voltage Differential Signal (LVDS) formats to transfer data into and out of an IC. These models are based on simplified circuit expressions that may include replacement of active circuit elements with ideal controlled voltage and current sources. As such, simulation with macromodels should be treated as 'typical' performance and not relied upon as final proof-of-design. Use of macromodel descriptions is not a substitute for worst-case design analysis, nor for testing real circuits over temperature, supply, and other operating limits.

The output format is provided as ASCII text netlists suitable for generic SPICE. This format is compatible with most versions of SPICE such as PSPICE and HSPICE. Additional information is found in HFAN 6.1 *Input/Output Models for Maxim Fiber Components*.

To extract the circuit netlists using the Adobe Acrobat Reader follow these instructions. Select the "Text Select Tool" by clicking the left mouse button on this icon of the menu bar (a capital T with a small dashed box to the lower right). Highlight the desired netlist text with the cursor. Use the copy command from the edit menu to capture the selected lines. Then paste the selected lines into a text file editor and save the file with an extension compatible with the simulator.

MAX3273 Low Power Laser Driver




Figure 2. Simplified input package model and input circuitry for the MAX3273

Version A, March 17 2003

Notes:

The schematic on the previous page represents the input stage of the Maxim MAX3273 Low Power Laser Driver. The output stage is not modeled since the electrical model is not considered an adequate representation for the analysis of the resultant optical signal.

The netlist is in SPICE 2g6 format. Since nodes in SPICE 2g6 can only be numbers, the output signals are 2101 and 2102 and the input signals to the die are 1001 and 1002. Comments in the netlist identify the correspondence between the signal names and the node numbers. The netlists are in SPICE 2g6 format and are compatible with PSPICE and HSPICE. It has been simulated on a generic SPICE simulator and PSPICE.

The Input Stage: The input stage is comprised of two subcircuits INPKG and DRV_IN.

Text File Format: This model is shipped in “pdf” format. Models and netlists can be copied to text format in the Acrobat Reader by holding the left mouse button on the “Text Select Tool.” Then the user can “select” what netlist and/or subcircuit with the mouse. Then use the copy command from the “edit” menu to capture the selected lines. These lines can then be “pasted” into the user’s text file.

Circuit Netlist – Input circuit

```
INPUT - MAX3273 INPUT CIRCUIT
*
* THIS IS THE INPUT OF THE MAX3273
* NOTE AS PER TYPICAL APPLICATION CIRCUIT A 115 OHM RESISTOR
* HAS BEEN PLACED ACCROSS THE INPUTS TO THE DEVICE
*
*
.OPTIONS ACCT NOMOD NOPAGE LIMPTS=10000 RELTOL=.001
.WIDTH OUT=80
.TEMP 27
*
* INPUT SIGNALS IN+ = 2101, IN- = 2102
.OP
.TRAN 5PS 2.5NS
VCC 101 0 DC 3.3
* The power supply is 3.3Volts.
*
VINA 2 0 PULSE (3.3 2.5 0.2N 0.100N 0.100N 0.8200N 1.880N)
VINB 3 0 PULSE (2.5 3.3 0.2N 0.100N 0.100N 0.820N 1.880N)
RLOAD1 2 2101 50
CLOAD1 101 2101 0.2P
RLOAD2 3 2102 50
CLOAD2 101 2102 0.2P
CLOAD3 2101 2102 0.05P
*
RSHUNT 2101 2102 100
XPK1 2101 2102 8 9 0 0 0 INPKG
*
XCIROUT 8 9 101 INBUF
*
.SUBCKT INBUF 1001 1002 101
*
*
XRND1 1001 10 0 RND125
XRND2 10 11 0 RND125

XRND3 1002 12 0 RND125
XRND4 12 13 0 RND125
XRCDBAR1 1001 4 0 RCDBAR_5K
XRCDBAR2 1001 5 0 RCDBAR_5K
R1 4 101 16k
R2 4 0 24k
R3 5 101 16k
R4 5 0 24k

XQ1 14 11 1 0 N102M_2
*Shorten model name from N102m010_2
XQ2 16 13 1 0 N102M_2
R5 101 13 2000
R6 101 16 2000
IB1 1 0 0.20M

XQ3 11 11 20 0 N102M_1
*Shorten model name from N102m010_1
XQ4 20 20 13 0 N102M_1
XQ5 13 13 21 0 N102M_1
XQ6 21 21 11 0 N102M_1
*
XPAD1 1001 101 0 0 PADES50
XPAD2 1002 101 0 0 PADES50
*
```

```

.ENDS INBUF
*
*
.SUBCKT INPKG 1101 1102 1201 1202 110 111 112
*
* resistors
*
R1 1201 1203 215M
R2 1202 1204 195M
*
* inductors
L1 1101 1203 0.865N
L2 1102 1204 0.849N
*
* capacitors
*
C01 1101 0 115F
C02 1102 0 113F
*
* mutual capacitors
*
C01_02 1101 1102 39.00F
.ENDS INPKG
*
*
*BEGIN PROCESS MODELS
*
*
.SUBCKT RND125 1 2 3
*Revision: 1.1 Date: 1999/10/25 16:56:08
*
R1 1 4 62.500 TC=150U
R2 4 2 62.500 TC=150U
C1 1 3 6.610E-016
C2 4 3 2.644F
C3 2 3 6.610E-016
*XREPORT1 0 REPORTERL1N7
*XREPORT2 0 REPORTERL1N8
.ENDS RND125

.SUBCKT RCDBAR_5K 1 2 3
*Revision: 1.1 Date: 1999/10/25 16:56:08

R1 1 4 2.513K
R2 4 2 2.513K
C1 1 3 2.352E-016
C2 4 3 9.410E-016
C3 2 3 2.352E-016
*XREPORT1 0 REPORTERL1N3
*XREPORT2 0 REPORTERL1N4
* read ~/lib/gst2/generic/models/resistors/inst_rcd
.ENDS RCDBAR_5K

.SUBCKT N102M_2 1 2 3 21

*.SUBCKT N102M010_2 1 2 3 21
CP1SUB 2 201 2.423F
RP1SUB 20 201 100K
CTRENCH 1 202 10.017F
RFIELDDEPI 202 21 500MEG
RREVERT 202 20 1M
CBL 10 20 1.380F
RSUB 20 21 309.654K
CWAFFER 20 21 1.701F

```

CP1EPI 10 12 1.523F
CP1P2 12 3 1.326F
RBX 2 12 117.486 TC=2.423M
RCX 1 10 43.405 TC=2.815M,503.797N
RCI 10 11 10.851 TC=2.815M,503.797N
REX 13 3 39.913
QN 11 12 13 11 TX 2
*XREPORT1 0 REPORTERL1N0
*XREPORT2 0 REPORTERL1N2
.MODEL TX NPN(IS=1.102E-018 XTI=3 EG=1.120 BF=380 BR=12 XTB=0 VAF=66
+ VAR=2.500 NF=1.018 NR=1.020 NE=2 NC=1.560 IKF=2.425M IKR=68.900U
+ ISE=5.512E-019 ISC=0 RB=58.743 RBM=44.057 IRB=248.040U CJE=2.619F
+ MJE=463M VJE=1.100 FC=990M CJC=1.747F MJC=350M VJC=1 TF=1.320P TR=5N
+ XTF=2 VTF=1.200 ITF=8.957M PTF=5 KF=545.257N AF=2.150)
.ENDS N102M_2

.SUBCKT N102M_1 1 2 3 21
CP1SUB 2 201 1.212F
RP1SUB 20 201 100K
CTRENCH 1 202 5.008F
RFIELDDEPI 202 21 1G
RREVERT 202 21 1M
CBL 10 20 6.901E-016
RSUB 20 21 619.308K
CWAFAFER 20 21 8.506E-016
CP1EPI 10 12 7.614E-016
CP1P2 12 3 6.631E-016
RBX 2 12 234.971 TC=2.423M
RCX 1 10 86.810 TC=2.815M,503.797N
RCI 10 11 21.703 TC=2.815M,503.797N
REX 13 3 79.826
QN 11 12 13 11 TX
*XREPORT1 0 REPORTERL1N0
*XREPORT2 0 REPORTERL1N1
.MODEL TX NPN(IS=1.102E-018 XTI=3 EG=1.120 BF=380 BR=12 XTB=0 VAF=66
+ VAR=2.500 NF=1.018 NR=1.020 NE=2 NC=1.560 IKF=2.425M IKR=68.900U
+ ISE=5.512E-019 ISC=0 RB=58.743 RBM=44.057 IRB=248.040U CJE=2.619F
+ MJE=463M VJE=1.100 FC=990M CJC=1.747F MJC=350M VJC=1 TF=1.320P TR=5N
+ XTF=2 VTF=1.200 ITF=8.957M PTF=5 KF=1.210U AF=2.150)
.ENDS N102M_1

.SUBCKT PADES50 2 3 4 5
XP1 2 5 PAD4SQ3P7
XQ1 2 3 5 DE0396
XQ2 4 2 5 DE0396
.ENDS PADES50

.SUBCKT DE0396 1 2 21
CTRENCH 2 202 33.018F
RFIELDDEPI 202 21 291.149
RREVERT 202 21 1G
CBL 4 5 37.842F
RSUB 5 21 49.887K
CWAFAFER 5 21 10.560F
CP1EPI 1 4 32.900F
DD 1 4 DCB
RS 4 2 12.010 TC=4.306M,4.262U
*XREPORT1 0 REPORTERL1N28
*XREPORT2 0 REPORTERL1N29
.MODEL DCB D(IS=3.485E-018 N=1.050 CJO=95.040F VJ=800M M=500M)
.ENDS DE0396

.SUBCKT PAD4SQ3P7 1 3
CPAD 1 10 67.534F
REPI 10 20 378.507 TC=4.800M,5U

```
CTRENCH 21 20 22.531F
CBL 21 20 1.413P
RX 20 21 1G
RS 3 21 5.416K
CWAFFER 21 3 2.587F
*XREPORT1 0 REPORTERL1N36
.ENDS PAD4SQ3P7

*.PRINT TRAN V(8) V(9)
.PRINT TRAN V(2101) V(2102)
*.PROBE
.END
```